
OPL Workshop 1b -  Einstieg für den Psion S5xx 
Ulrich Krinzner, 5/1998 

Wenn Sie sich mit OPL beschäftigen wollen, ist Ihnen sicherlich bereits klar, daß das Handbuch Ihr treuer 
Begleiter sein wird. Wir wollen Ihnen die Mühsal des Studierens nicht abnehmen, sondern durch unseren 
Beitrag auf Schwerpunkte hinweisen und Sie anregen, sich mit den starken Fähigkeiten Ihres auch sonst 
schon so nutzvollen Begleiters vertraut zu machen. Kleine selbstprogrammierte Anwendungen können das 
Leben noch viel angenehmer machen ... 

Viele Aspekte bezüglich der OPL-Programmierung , die wir Ihnen bereits in dem  entsprechenden Artikel für 
den S3a/c hier im Magazin geschildert habe, treffen unverändert auch auf den S5 zu. Lesen Sie bitte daher 
diesen Artikel ebenfalls, denn in einzelnen  Punkten setzen wir die Kenntnis des Inhaltes als bereits bekannt 
voraus. Wir möchten wegen der Systematik auch bei dem gleichen Schema und den Programmen wie dort 
bleiben.  
Generell ist das OPL-Programmieren auf dem S5 nicht anders als auf dem S3a/c. Und doch gibt es zum Teil 
erhebliche Sprachunterschiede - überwiegend wegen des Ziels, die speziellen Eigenschaften des Serie 5 
optimal auszunutzen. So sind einige Befehle hinzugekommen und andere weggefallen. Außerdem gibt es 
einige praktische Zusätze, mit denen wir uns heute noch nicht beschäftigen wollen. Weitere Veränderungen 
betreffen die Bildschirmgröße und die Möglichkeit, Grafiken in Graustufen darzustellen. Auch das wird uns 
erst in einem späteren Teil beschäftigen. 

Das erforderliche Handwerkszeug 

Anders als bei Serie 3a/c wird beim Verkauf des Serie 5 kein Programmierhandbuch dem Ausstattungspaket 
mehr beigelegt. Wer sich also mit OPL auseinandersetzen will oder muß, sollte sich zuerst mit diesem 
ausrüsten. Zur Zeit ist es über Psion PLC (UK) aus dem Internet zu beziehen. Die Quelle: http://
www.psion.com/downloads/docs/opl/. Aktuell sind dort gelagert:  
- OPLMAN32.ZIP (ca. 700kB, Programmierhandbuch als Winword-Dateien) 
- OPLGUIDE.ZIP (ca. 1,8 MB, Programmierhandbuch als Acrobat-Reader-PDF-File) 
- PORTING.ZIP (ca. 200kB, Informationen zur Portierung OPL16 zu OPL32, PDF-File) 
Immerhin hat man einen Stapel von nicht ganz 400 Seiten nach dem Ausdruck der Dokumentation vor der 
Nase. Das kommt zum Teil dadurch, daß es sich um ein Gesamtwerk für Siena, S3a/c und S5 handelt. 
Wer sich bereits mit OPL auf dem S3a/c auskennt, kann für den Anfang sicherlich etwas mit dem Porting 
Guide anfangen, der speziell auf die Unterschiede zwischen den beiden OPL-Versionen eingeht. 
Ausgedruckt ist das aber auch noch eine Menge Holz vor der Hütte... 
Ach ja, nicht zu vergessen: Einen Englisch-Kurs sollten Sie zum Lesen der Dokumente bereits hinter sich 
haben. 
Noch stehen für den Normal-Anwender nicht so viele hilfreiche Programme wie für den S3a/c zur Verfügung. 
Das Hauptwerkzeug wird wohl der ins Gerät eingebaute OPL-Editor sein. Schließlich ist er gerade auch für 
kurze Programme das effektivste Werkzeug, denn mit der S5-Tastatur ist die Eingabe von Text bei weitem 
nicht mehr so beschwerlich wie bei den Vorgängern. 
Dem ambitionierten Sharewareprogrammierer und dem Profi steht eine komfortable Alternative zur Auswahl: 
ein unter Windows 95 laufender Emulator. An dieses Programm kommt man allerdings erst, wenn man sich 
bei Psion Software als OPL-Entwickler "outet", und einen finanziellen Beitrag von etwa DM 200,- leistet. 
Dafür gibt es dann eine CD mit allerlei kleinen Helferchen und Informationen sowie den Zugang zu allen 
brandaktuellen Informationen rund um's OPL-Programmieren via Internet. Siehe im Internet: http://
www.software.psion.com/ bzw. http://developer.epocworld.com/. 

Sie können sich vielleicht denken, daß wir unter diesen Umständen beim eingebauten Editor bleiben ...  
Da Sie den Weg in diesen Artikel gefunden habe, gehen wir davon aus, daß Sie sich mit den Möglichkeiten 
des Menüsystem und der Symbolleiste recht schnell selbst vertraut machen können. Trotzdem seien ein 
paar Abweichungen zur Vorgängerversion erwähnt: 
- Der Editor ist kein reiner Editor mehr, er legt ein eigenes Datei-Format an. Durch die Import- und 
Exportfunktion ist es jedoch recht einfach, reine Texte (wie sie vom Editor auf dem S3a/c erzeugt wurden) 

einzufügen oder auch abzulegen. Wer also glaubt, passend zum Internetbrowser hier den Editor zur 
Erstellung von HTML-Seiten gefunden zu haben, der irrt. Leider. 
- Sonderzeichen sind wie unter Word recht einfach einzufügen. 
- Die Funktionen "Suchen", "Übersetzen" und "Gehe zu" sind schnell erreichbar auf die Symbolleiste gelegt - 
seeeehr praktisch! 
Ansonsten findet man alle nützlichen Funktionen des Psion Serie 5 auch im Editor wieder, z.B. Kopieren & 
Einfügen usw. 

Wiederverwendbarkeit von S3a/c-OPL-Programmen 

Die Frage der Wiederverwendbarkeit "alter" S3a/c-Programme wird gerade von "Normalnutzern" recht oft 
gestellt. Sie ist mit einem eindeutigen "Jein" zu beantworten.  
Übersetzte Module (*.OPA, *.OPO) fallen in die Rubrik "Nein". Die Ursache ist im Wechsel des 
Betriebssystems zu suchen.  
Als Programmierer sind Sie trotzdem nicht so übel dran, wenn Sie den Quelltext (*.OPL) Ihres Programmes 
aus der verstaubten Truhe auf dem Dachboden hervorkramen können. Der Quelltext liegt im Format "reiner 
Text" vor und läßt sich nach Importieren in den S5-Editor und erneuter Übersetzung wieder verwenden. 
Wenn es sich um schlichte Programme wie unsere Demos handelt, bekommen Sie das Baby recht schnell 
wieder zum Laufen. Ist dagegegen Bildschirmgestaltung/Grafik (und so manches andere "höherwertige") 
Bestandteil, sind unter Umständen einige Stunden der kostbaren Freizeit mit dem Lesen des Porting Guide 
und der Fehlersuche ans Bein zu binden. Um es positiv auszudrücken: Das enthaltene Lehr- und 
Lernstoffpotential ist recht hoch ... 

Praxis: Tastaturabfrage 

Zum Sammeln der ersten praktischen Erfahrungen sehen wir uns die Tastaturabfrage an, die letztendlich mit 
dazu dienen soll, dem Anwender immer einen "normalen" Programmausstieg zu erlauben (s. S3a/c-OPL 
Artikel). 
Der Programmrahmen "quest5" zeigt einen etwas geänderten Aufbau im Vergleich zur  S3a/c-Version. Das 
liegt an dem Einsatz des neuen Befehls GETEVENT32. Prinzipiell ist er mit dem alten GETEVENT 
vergleichbar, berücksichtigt aber nun auch die Ereignisse, die aus der Stiftbedienung resultieren. Dadurch 
ergibt sich auch eine andere Belegung des erweiterten Variablenfeldes, in dem die Ereignisse ihre Spuren in 
Form von Zahlen hinterlassen. Der Befehl TESTEVENT wurde deshalb weggelassen - auch weil er nach der 
Darstellung des Handbuches gerade in Schleifen zu einem erhöhten Stromverbrauch beiträgt. 

PROC quest5:  
  REM nur Programmprinzip, nicht lauffähig!  
  GLOBAL k&(16), a1&, a2&, esc%  
  esc%=27  
  DO  
    GETEVENT32 k&()  
    IF (k&(1) AND &400)=0 AND k&(5)=0  
      REM das Ereignis war ein einfacher Tastendruck  
      a1&=k&(1)   REM keycode  
      a2&=k&(4)   REM Modifizierertaste  
      IF a2&=0   REM keine Modifizierertaste gedrückt  
        IF a1&=esc%  
          BEEP 5,333  
          CLS  
          PRINT "Programmende, beliebige Taste drücken..."  
          GET  
          BREAK  
        ELSEIF a1&=4150   REM Menütaste  
          REM rufe Unterprogramm "Menü" auf  
        ENDIF  
     ELSEIF a2&=2   REM Shift gedrückt  
        REM weitere Auswahl  
     ELSEIF a2&=4   REM Strg gedrückt  
        IF a1&=8    REM "H"  
          REM rufe programmeigene Hilfe auf  
        ELSEIF a1&=5   REM "E"  
          REM rufe Programmabbruch auf  
        ELSEIF ... weitere Bedingungen  
          ...  
        ENDIF  
     ENDIF  
    ENDIF  
  UNTIL 0  
ENDP  

Das Programm stellt nicht für jeden Fall die Ideallösung dar, läßt aber das Prinzip erkennen.  
Einige Tasten sind auf dem S5 anders zugänglich oder weggefallen, so daß andere Lösungen erforderlich 
sind. So empfiehlt es sich nicht mehr, die Help-Taste (die einer Tastenkombination gewichen ist) direkt 
einzubeziehen, denn die umfangreiche Psion-eigene Hilfe sollte  sinnvollerweise ständig verfügbar sein. Die 
Folge: Man programmiert ein eigenes Hilfemenü mit selbständigem "Shortcut". 
Beachten Sie, daß auf dem Serie 5 die Bezeichnung für die Standard-Shortcuts geändert wurde. Statt des 
wohlvertrauten Psion-X für den "eXit" auf dem S3a/c lautet die neue Formel Strg-E  für "Exit". Die Portierung 
eines alten S3a/c-Programmes muß sowohl diesem Belang  als auch dem Wegfall der Psion-Taste 
Rechnung tragen! 

Welche Tastenkombinationen welche Variablenwerte erzeugen, ist Gegenstand des kurzen Programms 
"getevent". Beim Lesen des Programmtextes wird die Verwendung der Feldvariablen sichtbar, in denen bei 
Tastaturereignissen die entsprechend verwertbaren Daten abgelegt werden. Der erwähnte Scancode ist ein 
Wert, der die Belegung auf dem Tastaturfeld beschreibt. 

PROC getevent:  
  REM Ermittelt einige getevent32-Werte  
  REM für Tastatureingaben  
  GLOBAL k&(16), a1&, a2&  
  DO  
    GETEVENT32 k&()  
    IF (k&(1) And &400)=0 AND k&(5)=0  
      a1&=k&(1)  
        IF a1& <>27  
          beep 1,333  
        ELSE a1&=27  
          BEEP 1,444  
          CLS  
          PRINT "Ende des Programms"  
          GET  
          BREAK  
        ENDIF  
      PRINT "Tastenwert:", a1&  
      PRINT "Scancode:",k&(3)  
      PRINT "Modifizierertaste:",k&(4)  
      PRINT "Tastenwiederholungen:",k&(5)  
      PRINT  
    ENDIF  
  UNTIL 0  
ENDP  

Wenn Sie das Programm laufen lassen, betrachten Sie einmal aufmerksam das Ergebnis. Manche 
Tastaturkombinationen ergeben nicht nur eine Veränderung der Modifizierertaste, sondern beeinflussen auch 
den Tastenwert. Das kann die Programmierung sowohl erleichtern als auch erschweren. Auf alle Fälle lohnt 
es sich, die Informationen vor dem Programmieren zur Hand zu haben! 

Alternatives Menü 
Erste kleinereTücken stellen sich bereits bei der Übertragung unserer Menü-Alternative aus dem S3a/c-
Artikel heraus.  

!  
Abb.: Schlichtes Menü 

Bei simpler Übertragung des Quelltextes vom S3a/c auf den S5 war einfach der Balken, der beim S3a/c die 
Auswahl markierte, spurlos verschwunden.  Ursache: Die etwas andere Nutzung des dTEXT-Befehles. Ein 
Bindestrich (oder jeder andere Inhalt) sorgt dafür, daß alles wieder im Lot ist. Das Ergebnis (s.Abb.) zeigt, 
warum das alles gerade so und nicht anders ist. Der ehemalige Markierbalken ist jetzt nämlich auf den 
"Intro-String" begrenzt! 

PROC NewMenu5:  
  REM Programm nicht komplett, aber lauffähig  
  REM "UP" steht für "Unterprogramm"  
  LOCAL a%  
  dINIT "Menü Datei"  
  dTEXT "-", "Laden", $400  
  dTEXT "-", "Speichern", $600  
  dTEXT "-", "Beenden", $400  
  a%=DIALOG  
  PRINT a%  
  GET  
  IF a%=2  
    REM rufe UP "Laden"  
  ELSEIF a%=3  
    REM rufe UP "Speichern"  
  ELSEIF a%=4  
    REM rufe UP "Beenden"  
  ENDIF  
ENDP

Die "$600" sorgt weiterhin für eine Strich unter dieser Zeile. Die Endziffern, die die Positionierung des 
zweiten Strings beeinflussen (links, zentriert, rechts), haben aber keinen Einfluß, wenn der "Intro-String" mit 
einem Inhalt belegt ist. Demzufolge stehen hier in der Kennung "runde" Zahlenwerte. 

Temperaturkonverter 
Das Programm ist nahezu identisch mit dem S3a/c-Programm und würde auch in der alten Version laufen. 
Der klitzekleine Unterschied liegt in der Positionierung des zweiten Strings, worauf wir wegen der 
Gesamtkonstellation von dTEXT (wie gerade erwähnt) keinen Einfluß mehr haben. Also wurden die 
entsprechenden Werte verändert bzw. entfernt. 

!  
Abb.: Temperaturkonverter ala Serie 5 

PROC tconv5:  
  REM Umrechnung von Temperatureinheiten;  
  REM es gilt °C=(°F-32)/1,8  
  GLOBAL fz, c2f$(10), f2c$(10), c2f, f2c  
  c2f$=" " : f2c$=" "  
  DO  
    dINIT "Temperaturkonverter"  
    dFLOAT fz, "Temp. eingeben: ",-10000,10000  
    dTEXT "", "Ergebnisse(nur volle Grad):"  
    dTEXT "°C -->°F:", c2f$  
    dTEXT "°F -->°C:", f2c$,$200  
    dTEXT "", "Beenden: mit ESC-Taste!",2  
    IF DIALOG  
      c2f=32+(1.8*fz) : c2f$=FIX$(c2f,0,10)  
      f2c=(fz-32)/1.8 : f2c$=FIX$(f2c,0,10)  
    ELSE  
      BEEP 3,333 : STOP   REM Programmende  
    ENDIF  
  UNTIL 0  
ENDP  

Mit diesem Einstieg wollen wir Sie nun in Ruhe eigenen Versuchen überlassen.  


